Math 2650

A. J. Meir

We will now look at (linearity and at) linear second order differential equations.

```
> restart:with(DEtools):with(plots):
```

Consider the equation

$$\frac{d^2x}{dt^2} + a(t)\frac{dx}{dt} + b(t)x = g(t) \qquad x(0) = x_0, \quad \frac{dx}{dt}(0) = v_0.$$

This is the prototype linear, second order, differential equation and initial conditions. Together they form an initial value problem (i.v.p.).

We consider the differential equation (d.e.):

(*) $$\frac{d^2x}{dt^2} + a(t)\frac{dx}{dt} + b(t)x = g(t)$$

and the corresponding homogeneous d.e.

(**) $$\frac{d^2x}{dt^2} + a(t)\frac{dx}{dt} + b(t)x = 0$$

(note the zero r.h.s.).

Due to linearity if ϕ_1 and ϕ_2 are two solutions of the homogeneous d.e. (**) then also so is every linear combination of ϕ_1 and ϕ_2, i.e., $\phi(t) = c_1\,\phi_1(t) + c_2\,\phi_2(t)$, where c_1 and c_2 are arbitrary constants. That is for any constants c_1 and c_2, $\phi(t) = c_1\,\phi_1(t) + c_2\,\phi_2(t)$ is a solution of (**).

Also assume that ϕ_g is a solution of (*), (a particular solution), then due to linearity, for any constants c_1 and c_2, $\phi(t) = \phi_g(t) + c_1\,\phi_1(t) + c_2\,\phi_2(t)$ is a solution of (*).

Furthermore if ϕ_f is a solution of (*) with the r.h.s. $g(t)$ replaced by $f(t)$, then (due to linearity) $\phi(t) = \phi_g(t) + \phi_f(t)$ is a solution of (*) with the r.h.s. $g(t)$ replaced by $g(t) + f(t)$ and also for any constants c_1 and c_2, $\phi(t) = \phi_g(t) + \phi_f(t) + c_1\,\phi_1(t) + c_2\,\phi_2(t)$ is a solution of (*) with the r.h.s. $g(t)$ replaced by $g(t) + f(t)$.

Examples:
Lets define the differential equation:

```
> de:=D(D(x))(t)+a*D(x)(t)+b*x(t)=g(t);
```

$$de := \mathrm{D}^{(2)}(x)(t) + a\,\mathrm{D}(x)(t) + b\,x(t) = g(t)$$

```
> ic:=x(0)=x_0,D(x)(0)=v_0;
```
$$ic := x(0) = x\_0,\ \mathrm{D}(x)(0) = v\_0$$

We will now consider homogeneous equations (with $g(t) = 0$) and with constant coefficients (a and b constants). We will choose some values for a and b and solve the equation using the dsolve command. You can also try solving the equation by hand.

```
> g(t):=0;a:=0;b:=9;
```
$$g(t) := 0$$
$$a := 0$$
$$b := 9$$

```
> sol:=dsolve(de,x(t));
```
$$sol := x(t) = \_C1 \sin(3\,t) + \_C2 \cos(3\,t)$$

```
> g(t):=cos(t);
```
$$g(t) := \cos(t)$$

```
> sol:=dsolve(de,x(t));
```
$$sol := x(t) = \sin(3\,t)\ \_C2 + \cos(3\,t)\ \_C1 + \frac{1}{8}\cos(t)$$

Observe the homogeneous solution is $x_h := c_1 \sin(3\,t) + c_2 \cos(3\,t)$ and the particular solution is $x_p = \frac{\cos(t)}{8}$.

```
> g(t):=0;
```
$$g(t) := 0$$

```
> x_0:=1;v_0:=0;
```
$$x\_0 := 1$$
$$v\_0 := 0$$

```
> de;ic;
```
$$\mathrm{D}^{(2)}(x)(t) + 9\,x(t) = 0$$
$$x(0) = 1,\ \mathrm{D}(x)(0) = 0$$

```
> sol1:=dsolve({de,ic},x(t));
```
$$sol1 := x(t) = \cos(3\,t)$$

```
> sol1:=rhs(sol1);
```
$$sol1 := \cos(3\,t)$$

```
> plot(sol1,t=0..4);
```




```
> g(t):=sin(2*t);
```

$$g(t) := \sin(2\,t)$$

```
> de;ic;
```

$$D^{(2)}(x)(t) + 9\,x(t) = \sin(2\,t)$$

$$x(0) = 1, D(x)(0) = 0$$

```
> sol2:=dsolve({de,ic},x(t));
```

$$sol2 := x(t) = -\frac{2}{15}\sin(3\,t) + \cos(3\,t) + \frac{1}{5}\sin(2\,t)$$

```
> sol2r:=rhs(sol2);
```

$$sol2r := -\frac{2}{15}\sin(3\,t) + \cos(3\,t) + \frac{1}{5}\sin(2\,t)$$

```
> plot(sol2r,t=0..13);
```




```
> xh:=select(has,sol2r,3*t);
```

$$xh := -\frac{2}{15}\sin(3\,t) + \cos(3\,t)$$

```
> xp:=select(has,sol2r,2*t);
```

$$xp := \frac{1}{5}\sin(2\,t)$$

```
> odetest(x(t)=cos(3*t)-2/15*sin(3*t),de);
```

$$-\sin(2\,t)$$

```
> odetest(x(t)=1/5*sin(2*t),de);
```

$$0$$

```
> g(t):=sin(3*t);
```

$$g(t) := \sin(3\,t)$$

```
> x_0:=1;v_0:=0;
```

$$x\_0 := 1$$

$$v\_0 := 0$$

```
> de;ic;
```

$$D^{(2)}(x)(t) + 9\,x(t) = \sin(3\,t)$$

$$x(0) = 1, D(x)(0) = 0$$

```
> sol3:=dsolve({de,ic},x(t));
```

$$sol3 := x(t) = \frac{1}{18}\sin(3\,t) + \cos(3\,t) - \frac{1}{6}\cos(3\,t)\,t$$

```
> sol3r:=rhs(sol3);
```

$$sol3r := \frac{1}{18}\sin(3\,t) + \cos(3\,t) - \frac{1}{6}\cos(3\,t)\,t$$

```
> plot(sol3r,t=0..13);
```




```
> plot(sol3r,t=0..20);
```



```
> g(t):=sin(2*t);
```

$$g(t) := \sin(2\,t)$$

```
> sol4:=dsolve(de,x(t));
```

$$sol4 := x(t) = \sin(3\,t)\ \_C2 + \cos(3\,t)\ \_C1 + \frac{1}{5}\sin(2\,t)$$

```
> sol4:=rhs(sol4);
```

$$sol4 := \sin(3\,t)\ \_C2 + \cos(3\,t)\ \_C1 + \frac{1}{5}\sin(2\,t)$$

```
> xh:=select(has,sol4,3*t);
```

$$xh := \sin(3\,t)\ \_C2 + \cos(3\,t)\ \_C1$$

```
> xp:=select(has,sol4,2*t);
```

$$xp := \frac{1}{5}\sin(2\,t)$$

```
> x_0:='x_0';v_0:='v_0';
```

$$x\_0 := x\_0$$

$$v\_0 := v\_0$$

```
> eq1:=subs(t=0,xh+xp)=x_0;
```

$$eq1 := \sin(0)\ \_C2 + \cos(0)\ \_C1 + \frac{1}{5}\sin(0) = x\_0$$

```
> eq2:=subs(t=0,diff((xh+xp),t))=v_0;
```

$$eq2 := 3\ \_C2\cos(0) - 3\ \_C1\sin(0) + \frac{2}{5}\cos(0) = v\_0$$

```
> eval(eq1);
```

$$\_C1 = x\_0$$

```
> eval(eq2);
```

$$3\ \_C2 + \frac{2}{5} = v\_0$$

Exercise:

Solve the equation:

$$\frac{d^2 x}{dt^2} + 16\,x = Sin(2\,t) \qquad x(0) = 0, \quad \frac{dx}{dt}(0)=1.$$

Identify the homogeneous solution and the particular solution.

Is the solution periodic, if it is what is the period?

Repeat the above for

$$\frac{d^2 x}{dt^2} + 16\,x = Sin(4\,t) \qquad x(0) = 0, \quad \frac{dx}{dt}(0)=1.$$